


Example of Fourier expansion

```
> f := unapply( piecewise( t < Pi, t,
                          t >= Pi, 2*Pi-t ), t );
      f := t → piecewise( t < π, t, π ≤ t, 2π - t)
```

(1)

```
> plot( f(t), t=0..2*Pi) ;
```



```
> a0 := int( f(t), t=0..2*Pi) / Pi ;
      a0 := π
```

(2)

```
> a1 := int( f(t)*cos(t), t=0..2*Pi) / Pi ;
      b1 := int( f(t)*sin(t), t=0..2*Pi) / Pi ;
```

$$a1 := -\frac{4}{\pi}$$

$$b1 := 0$$

(3)

```
> a2 := int( f(t)*cos(2*t), t=0..2*Pi) / Pi ;
      b2 := int( f(t)*sin(2*t), t=0..2*Pi) / Pi ;
```

$$a2 := 0$$

$$b2 := 0$$

(4)

```
> a3 := int( f(t)*cos(3*t), t=0..2*Pi) / Pi ;  
b3 := int( f(t)*sin(3*t), t=0..2*Pi) / Pi ;
```

$$a3 := -\frac{4}{9\pi}$$

$$b3 := 0$$

(5)

```
> a4 := int( f(t)*cos(4*t), t=0..2*Pi) / Pi ;  
b4 := int( f(t)*sin(4*t), t=0..2*Pi) / Pi ;
```

$$a4 := 0$$

$$b4 := 0$$

(6)


```
> a5 := int( f(t)*cos(5*t), t=0..2*Pi) / Pi ;  
b5 := int( f(t)*sin(5*t), t=0..2*Pi) / Pi ;
```

$$a5 := -\frac{4}{25\pi}$$

$$b5 := 0$$

(7)

```
> plot( [f(t), a0/2,  
        a0/2+a1*cos(t),  
        a0/2+a1*cos(t)+a3*cos(3*t),  
        a0/2+a1*cos(t)+a3*cos(3*t)+a5*cos(5*t)], t=0..2*Pi) ;
```



```
> ak := int( f(t)*cos(k*t), t=0..2*Pi) / Pi ;
```

$$ak := \frac{\frac{\sin(\pi k) k \pi + \cos(\pi k) - 1}{k^2} - \frac{\sin(\pi k) k \pi + 2 \cos(\pi k)^2 - \cos(\pi k) - 1}{k^2}}{\pi} \quad (8)$$

```
> ak := simplify(ak) assuming k::integer ;
```

(9)

$$ak := \frac{2(-1 + (-1)^k)}{k^2 \pi} \quad (9)$$

```
> ak := simplify(subs( k=2*q+1, ak)) assuming q::integer ;
```

$$ak := -\frac{4}{(2q+1)^2 \pi} \quad (10)$$

```
> Sum( ak*cos((2*q+1)*t), q=0..infinity) ;
```

$$\sum_{q=0}^{\infty} \left(-\frac{4 \cos((2q+1)t)}{(2q+1)^2 \pi} \right) \quad (11)$$

```
> plot( sum( ak*cos((2*q+1)*t), q=0..5 ), t=-2*Pi..2*Pi) ;
```

